# TP 1 : Des études de suites

## 1 Quatre vitesses de convergence

### Suite u

```
> u:=n->if n=0 then 1. else a:=u(n-1):a-a^2/2 fi:
Warning, ‘a‘ is implicitly declared local to procedure ‘u‘
```

Deux remarques : "1." au lieu de "1" pour avoir des flottants plutot que des fractions; stockage de u(n-1) dans a pour ne pas le calculer deux fois

```
> seq(u(k),k=0..10);u(100);u(1000);u(10000);
```

1., .5000000000, .3750000000, .3046875000, .2582702637, .2249184991, .1996243335, .1796993962, .1635534597, .1501785926, .1389017878

.01879155975

.001982781726

```
Error, (in u) too many levels of recursion
```

Il faudrait le calculer avec une boucle !

```
> u:=proc(n)
  local r,k;
  r:=1.:
  for k from 1 to n do r:=r-r^2/2 od:
  RETURN(r)
  end:
> seq(u(k),k=0..10);u(100);u(1000);u(10000);
```

1., .5000000000, .3750000000, .3046875000, .2582702637, .2249184991, .1996243335, .1796993962, .1635534597, .1501785926, .1389017878

.01879155975

.001982781726

.0001997806530

```
> seq(1/u(10^k),k=1..5);
```

7.199331383, 53.21538038, 504.3419489, 5005.489696, 50006.64011

```
> seq(ln(u(10^k)),k=1..5);
```

-1.973988158, -3.974347459, -6.223254508, -8.518290528, -10.81991108

```
> seq([%][k]-[%][k+1],k=1..4);
```

2.000359301, 2.248907049, 2.295036020, 2.301620552

```
> ln(10.);
```

2.302585093

Connaissez-vous une fonction simpe telle que ln(f(10x))=ln(f(x))-ln(10) ? Moi oui...

Bonus : on demande à Maple un développement asymptotique (le premier terme est un équivalent).

```
> asympt(rsolve({s(0)=1,s(n+1)=s(n)-s(n)^2/2},s(n)),n);
```

$$2\,\frac{1}{n}+\frac{\_C-2\ln(n)}{n^2}+\frac{1+\_C+\frac{1}{2}\_C^2-(2\,\_C+2)\ln(n)+2\ln(n)^2}{n^3}+\Bigg($$

$$\frac{5}{3}+\frac{5}{2}\_C+\frac{5}{4}\_C^2+\frac{1}{4}\_C^3-\left(5+5\ \_C+\frac{3}{2}\_C^2\right)\ln(n)+(5+3\ \_C)\ln(n)^2-2\ln(n)^3\Bigg/$$

$$n^4+\mathrm{O}\left(\frac{1}{n^5}\right)$$

## Suite v

```
> v:=proc(n)
  local r,k;
  r:=1.:
  for k from 1 to n do r:=arctan(r) od:
  RETURN(r)
  end:
> seq(v(k),k=0..10);v(100);v(1000);v(10000);
```

1., .7853981634, .6657737500, .5873841757, .5310915102, .4882103378,
.4541714734, .4263175062, .4029860576, .3830779112, .3658337989

.1219393604

.03871946551

.01224733212

```
> seq(1/v(10^k),k=1..5);
```

2.733481715, 8.200797484, 25.82680279, 81.65043539, 258.1986888

```
> seq(ln(v(10^k)),k=1..5);
```

-1.005576150, -2.104231404, -3.251412821, -4.402447152, -5.553729400

```
> seq([%][k]-[%][k+1],k=1..4);
```

1.098655254, 1.147181417, 1.151034331, 1.151282248

```
> ln(10.)/2;
```

1.151292546

```
> asympt(rsolve({s(0)=1,s(n+1)=arctan(s(n))},s(n)),n);
```

$$\frac{1}{2}\sqrt{3}\sqrt{2}\sqrt{\frac{1}{n}}+\left(\_C+\frac{3}{80}\sqrt{3}\sqrt{2}\ln(n)\right)\left(\frac{1}{n}\right)^{(3/2)}+\mathrm{O}\left(\frac{1}{n^2}\right)$$

```
> 2/sqrt(6.);
```

.8164965808

## Suite w

```
> w:=proc(n)
  local r,k;
  r:=1.:
  for k from 1 to n do r:=sin(r)/2 od:
  RETURN(r)
  end:
> seq(w(k),k=0..10);w(100);w(1000);w(10000);
```

1., .4207354924, .2042159564, .1013997354, .05061303045, .02529571208,
.01264650724, .006323085070, .003161521468, .001580758100, .0007903787210

$.6384627575\ 10^{-30}$

$.7553347655\ 10^{-301}$

$.4056752320\ 10^{-3010}$

```
> seq(1/w(10^k),k=1..5);
```

1265.216248, $.1566262070\ 10^{31}$, $.1323916289\ 10^{302}$, $.2465026014\ 10^{3011}$, $.1234329961\ 10^{30104}$

```
> seq(ln(w(10^k)),k=1..5);
```

-7.142998334, -69.52624472, -693.3587072, -6931.683332, -69314.92958

```
> seq([%][k+1]/[%][k],k=1..4);
```

9.733481861, 9.972618397, 9.997254322, 9.999725357

Cherchons une fonction telle que ln(f(10x))=10ln(f(x))...

```
> asympt(rsolve({s(0)=1,s(n+1)=sin(s(n))/2},s(n)),n);
```

$$\_C\left(\frac{1}{2}\right)^n+\frac{2}{9}\_C^3\left(\left(\frac{1}{2}\right)^n\right)^3+\frac{74}{675}\_C^5\left(\left(\frac{1}{2}\right)^n\right)^5+\mathrm{O}\left(\_C^6\left(\left(\frac{1}{2}\right)^n\right)^6\right)$$

Hop !

### suite z

```
> z:=proc(n)
  local r,k;
  r:=1.:
  for k from 1 to n do r:=sin(r)^2 od:
  RETURN(r)
  end:
> seq(z(k),k=0..10);z(100);z(1000);z(10000);
```

1., .7080734183, .4229831090, .1684958652, .02812319136, .0007907053995, $.6252148985\ 10^{-6}$, $.3908936693\ 10^{-12}$, $.1527978607\ 10^{-24}$, $.2334718623\ 10^{-49}$, $.5450911049\ 10^{-99}$

0.

0.

0.

```
> seq(1/z(k),k=1..5);
```

1.412282927, 2.364160598, 5.934863736, 35.55784218, 1264.693526

```
> seq(ln(z(k)),k=1..5);
```

-.3452074925, -.8604230322, -1.780844068, -3.571160728, -7.142585100

```
> seq([%][k+1]/[%][k],k=1..4);
```

2.492480757, 2.069730820, 2.005319159, 2.000073826

Cherchons une fonction telle que ln(f(x+1))=2ln(f(x))...

```
> asympt(rsolve({s(0)=1,s(n+1)=sin(s(n))^2},s(n)),n);
Error, (in asympt) unable to compute series
```

Dommage...

# 2 Un phénomène d'instabilité numérique

```
> restart;
```

## 1. Suite de Fibonacci

### *Formellement*

```
> rsolve({f(n+2)=f(n)+f(n+1),f(0)=0,f(1)=1},f(n));
```

$$\frac{\left(-1+\frac{1}{5}\sqrt{5}\right)\left(-2\,\frac{1}{1-\sqrt{5}}\right)^{n}}{1-\sqrt{5}}+\frac{\left(-\frac{1}{5}\sqrt{5}-1\right)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^{n}}{1+\sqrt{5}}$$

```
> subs(n=100,%);
```

$$1267650600228229401496703205376\,\frac{-1+\frac{1}{5}\sqrt{5}}{(1-\sqrt{5})^{101}}+\frac{1267650600228229401496703205376\left(-\frac{1}{5}\sqrt{5}-1\right)}{(1+\sqrt{5})^{101}}$$

```
> simplify(%);
```

$$22768695390444835768322853834430573808713702819118234838940828401158469061705728 00\,\frac{1}{(-1+\sqrt{5})^{101}\,(1+\sqrt{5})^{101}}$$

```
> numer(%)/expand(denom(%));
```

$$354224848179261915075$$

Je n’ai pas réussi à faire autrement ! Je voulais éviter le evalf...

*Numériquement*

```
> f:=proc(n)
  option remember;
  if n<=1 then n else f(n-1)+f(n-2) fi;
  end:
> f(100);
```

$$354224848179261915075$$

## 2. Suite g

*Formellement*

```
> rsolve({g(n+2)=g(n)+3/2*g(n+1),g(0)=1,g(1)=-2},g(n));
```

$$\frac{8}{5}\left(\frac{-1}{2}\right)^{n}-\frac{3}{5}\,2^{n}$$

```
> subs(n=100,%);
```

$$\frac{-1205203533194242706656471569255871951891652245337094626476 03}{1584563250285286751870879006 72}$$

```
> evalf(%);
```

$$-.7605903601\;10^{30}$$

*Numériquement*

```
> g:=proc(n)
  option remember;
  if n=0 then 1. elif n=1 then -2 else g(n-2)+1.5*g(n-1)
  fi;
  end:
> g(100);
```

$$-.7605903604\ 10^{30}$$

Tout va bien

# 3. Suite h

### *Formellement*

```
> rsolve({h(n+2)=h(n)+3/2*h(n+1),h(0)=-3,h(1)=3/2},h(n));
```

$$-3\left(\frac{-1}{2}\right)^n$$

```
> subs(n=100,%);
```

$$\frac{-3}{1267650600228229401496703205376}$$

```
> evalf(%);
```

$$-.2366582716\ 10^{-29}$$

### *Numériquement*

```
> h:=proc(n)
  option remember;
  if n=0 then -3. elif n=1 then 3/2 else h(n-2)+1.5*h(n-1)
  fi;
  end;:
```

$h$ := **proc**($n$)

**option** *remember*;

**if** $n = 0$ **then** -3. **elif** $n = 1$ **then** $3 / 2$ **else** $\mathrm{h}(n-2) + 1.5 * \mathrm{h}(n-1)$ **end if**

**end proc**

```
> h(100);
```

$$.6189688107\ 10^{14}$$

Héhé !!

```
> rsolve({h(n+2)=h(n)+3/2*h(n+1),h(0)=-3,h(1)=3/2},h(n));
Error, (in h) too many levels of recursion
```

Normal, h a maintenant une valeur... il faut changer de lettre, ou bien "réinitialiser" h. C’est ce choix que je fais.

```
> unassign(h):
> rsolve({h(n+2)=h(n)+3/2*h(n+1),h(0)=a,h(1)=b},h(n));
```

$$\frac{1}{2}\left(\frac{8}{5}a-\frac{4}{5}b\right)\left(\frac{-1}{2}\right)^n-\left(-\frac{1}{5}a-\frac{2}{5}b\right)2^n$$

Si la condition h(0)=-2h(1) est vérifiée, on a une suite géométrique de raison de valeur absolue <1, donc qui tend vers 0. Mais si cette condition n’est pas vérifiée, meme avec h(0)+2h(1) très faible, la puissance de 2 va faire exploser la solution. Si on fait un calcul numérique, les erreurs d’arrondis font que, meme si au départ h(0)+2h(1)=0, on va avoir h(N)+2h(N+1) très légèrement différent de 0 à un moment donné, et à partir de là, la composante géométrique de raison 2 est non nulle, et fait tout exploser...

# 4. Une suite de Fibonacci instable

```
> rsolve({i(n+2)=i(n)+i(n+1),i(0)=a,i(1)=b},i(n));
```

$$-\frac{1}{10}\,\frac{(3\sqrt{5}-5)\,(b+2\,a+\sqrt{5}\;b)\left(-2\,\frac{1}{1-\sqrt{5}}\right)^{n}}{1-\sqrt{5}}$$

$$+\frac{\frac{1}{10}\,(3\sqrt{5}+5)\,(b+2\,a-\sqrt{5}\;b)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^{n}}{1+\sqrt{5}}$$

C'est le premier terme qui a une raison >1. On va donc faire en sorte d'annuler au départ cette composante.

```
> rsolve({i(n+2)=i(n)+i(n+1),i(0)=-(1+sqrt(5))/2,i(1)=1},i(n
  ));
```

$$\frac{(-\sqrt{5}-3)\left(-2\,\frac{1}{1+\sqrt{5}}\right)^{n}}{1+\sqrt{5}}$$

```
> evalf(subs(n=100,%));
```

$$-.2042789460\;10^{-20}$$

```
> i:=proc(n)
  option remember;
  if n=0 then -(1+sqrt(5.))/2 elif n=1 then 1. else
  i(n-1)+i(n-2) fi;
  end:
> i(100);
```

$$-.5475376912\;10^{11}$$

BOUM

# 3 Des récurrences du premier ordre non monotone

## 1. u[n+1]=exp(-u[n])

```
> f:=x->exp(-x):plot({f(x),x},x=0..1);
```




```
> fsolve(f(x)=x);
```

.5671432904

```
> u:=n->if n=0 then 1. else f(u(n-1)) fi:
> seq(u(k),k=0..20);
```

1., .3678794412, .6922006275, .5004735006, .6062435351, .5453957860, .5796123355, .5601154614, .5711431151, .5648793474, .5684287250, .5664147332, .5675566373, .5669089119, .5672762322, .5670678984, .5671860501, .5671190401, .5671570440, .5671354902, .5671477143

Les suites extraites d'indices pairs et impairs semblent adjacentes.
Déjà, elles vérifient des relations de la forme v[n+1]=f(f(v[n])), et fof est strictement croissante, donc la relation v[1]>v[0] se propage : v est donc strictement croissante, et de meme w est strictement décroissante. Elles sont par ailleurs localisées dans [0,1], d'où les convergences. Leurs limites l1 et l2 sont solutions de l'équation f(f(x))=x. Bien entendu, l'unique X tel que f(X)=X vérifie également f(f(X))=X, mais il pourrait y en avoir d'autres, a priori.

```
> plot({f(f(x)),x},x=0..1);
```




Il semblerait qu'il n'y ait qu'un point fixe. Pour cela, on considère la différence f(f(x))-x

```
> diff(f(f(x))-x,x);
```

$$\mathbf{e}^{(-x)}\,\mathbf{e}^{(-\mathbf{e}^{(-x)})}-1$$

g:x->f(f(x))-x est strictement décroissante, et g(0)>0>g(1), donc il existe un unique point l tel que g(l)=0, c'est-à-dire f(f(l))=l; gagné.

## 2. u[n+1]=exp(-3u[n])

```
> g:=x->exp(-3*x):plot({g(x),x},x=0..1);
```




```
> fsolve(g(x)=x);
```

$$.3499696317$$

```
> v:=n->if n=0 then 1. else g(v(n-1)) fi:
> seq(v(k),k=0..20);
```

1., .04978706837, .8612579679, .07548857913, .7973466610, .09144295141, .7600820780, .1022590240, .7358146163, .1099814186, .7189638104, .1156841749, .7067682045, .1199950708, .6976866431, .1233092404, .6907842579, .1258892441, .6854582185, .1279168730, .6813013096

Les suites extraites d'indices pairs et impairs semblent respectivement croissante et décroissante (on le prouverait comme pour u) et localisées dans [0,1], d'où les convergences. Par contre, il semblerait que leurs limites l1 et l2 soient différentes. Elles sont solutions de l'équation g(g(x))=x. Ici encore, l'unique X tel que g(X)=X vérifie également g(g(X))=X, mais on va voir qu'il y a d'autres solutions.

```
> plot({g(g(x)),x},x=0..1);
```




```
> plot(g(g(x))-x,x=0..1);
```




C'est plus clair, non ?
Il semblerait qu'il n'y ait qu'un point fixe. Pour cela, on considère la différence f(f(x))-x

```
> fsolve(g(g(x))=x);
```

.1361198833

```
> fsolve(g(g(x))=x,x=.2..0.5);
```

.3499696317

```
> fsolve(g(g(x))=x,x=.5..1);
```

.6647397623

Une étude de x->gog(x)-x etablirait effectivement l'existence de trois solutions pour l'équation g(g(l))=l. Si on note x1,x2,x3 ces solutions (dans l'ordre croissant), on montrerait que l'intervalle [0,x1] est stable par gog, donc tous les termes pairs de v sont dans cet intervalle, donc majorés par x1, donc leur limite également, donc cette limite EST x1. De meme, on montrerait que les termes impairs tendent vers x3.